



AMIGOS DEL
MSX

AÑO 1 • N.º 2




295
pesetas

**LOS MEJORES
PROGRAMAS LISTADOS**

MSX

**15 PROGRAMAS PARA
TU ORDENADOR MSX**






LO ÚNICO...

UN CHORRO

EFICAZ

PARA UNA BUENA CARGA
DEL PROGRAMA EN TU ORDE-
NADOR, COMPRUEBE
QUE LA CABEZA DE SU
PLATINA ESTE
LIMPIA Y EL
AZIMUTH PERFECTA-
MENTE AJUSTADO.
LIMPIELA CON EL
PRODUCTO
SKUA.



LIMPIADOR DE RADIO CASSETTES



FABRICADO POR

ASKLE
CHEMICAL S.A.

ALICANTE (SPAIN)


BOLETIN DE PEDIDO

Nombre y Apellidos
Direccion
Ciudad Telf. D.P.

Deseo recibir..... Limpiadores de cassettes al precio de 420 Ptas. Unidad.
Forma de pago: ☐ Talón ☐ Contrareembolso ☐ Giro postal

Remitir a GTS, S. A. - Bailén, n.º 20 - 1.º Izda. D.P. 28005

AMIGOS DEL MSX



EDITORIAL


Tras un breve lapsus de tiempo, debido principalmente a las vacaciones, aquí estamos de nuevo con nuevos e interesantes programas para todos aquellos "AMIGOS DEL MSX".

Ante todo, comenzar agradeciendo el cariño con el que hemos sido recibidos por todos vosotros, y aunque se haya dicho muchas veces esto nos obliga a superarnos en ofrecer lo mejor cada mes.

En este ya nuestro número 2, os ofrecemos una serie de programas listados con mayor o menor dificultad, en los cuales podéis introducir una serie de cambios y mejoras, dejando de esta forma campo abierto a vuestra imaginación y vuestros conocimientos del ordenador.

Esperamos que os sean de utilidad e interés y, con vuestra voluntad y nuestra pequeña contribución, seáis capaces de elaborar poco a poco, programas cada vez más complicados.


Gracias y hasta el próximo mes.



Edita: Editorial GTS. C/ Bailén, 20. 1.º Izda. 28005 MADRID.
Secretaría Redacción: N. Vera Clavijo. Colaboradores:
Eugenio Garrido, J. F. Martínez. J. Bernal, R. Carralón, J.
Ramos, Juan Jesús Ortega. Dirección Artística y Técnica:
Jesús Negrete. Publicidad: Dpto. propio Avda. Medi-
terráneo, 41, 1.º C. 28007 MADRID. Fotocomposición: Gráf.
FUTURA, Sdad. Coop. Ltda. Imprime: Gráf. FUTURA, Sdad.
Coop. Ltda. Distribuye: R.B.A. Promotora de Ediciones,
S. A. Trav. de Gracia, 56. Atico 1.º. Teléfono: 200 82 56. De-
pósito Legal: M. 3988-1986

SUMARIO

EXPLICACION DE LOS PROGRAMAS	4
BIBLIOGRAFIA DEL MSX	34



EXPLICACION DE LOS PROGRAMAS

BATALLA

```
10 '*****
20 '****      ****
30 '****  BATALLA  ****
40 '****      ****
50 '*****
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 '      1986
100 '*****
110 '
120 COLOR 6,4,14
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(20,80),10:PRINT#1,"BATALLA"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,1,1
190 SCREEN 2, 2 :FOR K=1 TO 2 :B$="":FOR I=1 TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$
)):NEXT:SPRITE$(K)=B$:NEXT
200 B$="":FOR I=1 TO 3:READ A:B$=B$+CHR$(A):NEXT
210 SPRITE$(3)=B$
220 LINE(0,18)-(255,0),10,BF
230 OPEN "grp:"AS #1
240 VDP(1)=226
250 Z=RND(-TIME)
260 DIM A(40)
```



```

270 LINE(0,19)-(255,192),1,BF
280 N=RND(-TIME)
290 FOR EX=1 TO 255 STEP 2
300 EY=20+INT(RND(1)*170)
310 PSET(EX,EY),10
320 NEXT
330 CIRCLE(30,40),10,14
340 PAINT(30,40),14
350 CIRCLE(35,40),10,1
360 PAINT(35,40),1
370 Z=INT(RND(1)*80)+180
380 Z1=0:Z2=0:V1=0:V2=0
390 FOR I=1 TO 15
400 A(I)=INT(RND(1)*130)
410 IF A<A(I) THEN A=A(I)
420 NEXT
430 V1=150:Z2=25
440 FOR I=1 TO 15:CV=(55*A(I))/A:V2=190-CV
450 LINE(Z1,V1)-(Z2,V2),15
460 V1=V2:Z1=Z2
470 Z2=Z2+Z/10
480 NEXT
490 PAINT(9,191),15
500 FOR I=190 TO 130 STEP-1
510 IF POINT(15,I)<>15 AND POINT(31,I)<>15 THEN CY=I-16:GOTO530
520 NEXT
530 FOR I=190 TO 130 STEP-1
540 IF POINT(225,I)<>15 AND POINT(241,I)<>15 THEN FY=I-16:GOTO560
550 NEXT
560 LINE(13,CY+17)-(30,190),15,BF
570 LINE(223,FY+17)-(240,190),15,BF
580 PUTSPRITE1,(15,CY),10,1
590 PUTSPRITE2,(225,FY),2,2
600 PSET(5,5),10:PRINT#1,"ANGULO="
610 PSET(155,5),10:PRINT#1,"FUERZA="
620 Q$="1":B$="2"
630 PSET(110,5),10:PRINT#1,Q$
640 A$=INKEY$
650 IF A$="+" THEN ANG=ANG+1:LINE(60,5)-(90,13),10,BF
660 IF A$="-" THEN ANG=ANG-1:LINE(60,5)-(90,13),10,BF
670 IF ANG<0 THEN ANG=0
680 IF ANG>90 THEN ANG=90
690 IF A$=CHR$(13) THEN GOTO 720
700 PSET(60,5),10:PRINT#1,ANG
710 GOTO 640
720 A$=INKEY$
730 IF A$="+" THEN FU=FU+1:LINE(211,5)-(250,13),10,BF
740 IF A$="-" THEN FU=FU-1:LINE(211,5)-(250,13),10,BF
750 IF FU<100 THEN FU=100
760 IF FU>250 THEN FU=250
770 IF A$=CHR$(13) THEN GOTO 800
780 PSET(211,5),10:PRINT#1,FU
790 GOTO 720
800 F=ANG*.0174533
810 IF Q$="1" THEN R=CY-4 ELSE R=FY-4
820 FI=FU*2
830 E=E+4
840 IF E>FU THEN M=M+2
850 Y=INT(SIN(F)*E)-M*3
860 X=INT(COS(F)*E)
870 IF Q$="1" THEN BX=X+15:A=3 ELSE A=-3:BX=230-X
880 BY=R-Y
890 IF POINT(BX+A,BY+2)=15 THEN BO=1


```




```

900 IF BX>255 OR BX<0 THEN BO=1
910 PUTSPRITE 3,(BX,BY),6,3
920 IF BO=1 THEN 960
930 ON SPRITE GOSUB 1050
940 SPRITE ON
950 GOTO 830
960 BEEP:BEEP:CIRCLE(BX,BY),5,1:CIRCLE(BX,BY),3,1
970 PUTSPRITE 3,(BX,209),6,3
980 LINE(60,5)-(90,13),10,BF
990 LINE(110,5)-(125,13),10,BF
1000 LINE(211,5)-(250,13),10,BF
1010 SWAP Q$,B$
1020 FU=0:ANG=0:E=0:M=0:BO=0
1030 GOTO 630
1040 DATA 0,15,1F,11,11,15,15,B1,EB,85,81,89,A0,82,88,80,1C,18,10,10,36,3E,22,2A,
2A,22,6B,D5,1,D1,C5,C1,2A,36,22,22,2A,2A,23,20,20,AB,C2,80,88,81,81,91,10,38,7C,
C6,54,54,D4,84,85,7,11,1,81,D1,C1,C5,64,224,64
1050 I=0
1060 PLAY"T138M101CDCC","T138M101CCED","T110M101DDC"
1070 I=I+1:IF I>15 THEN 1140
1080 IF BX>150 THEN 1110
1090 PUTSPRITE1,(15,CY),1,1
1100 IF BX<150 THEN 1120
1110 PUTSPRITE 2,(225,FY),1,2
1120 FOR J=1 TO 20:NEXT
1130 GOTO 1070
1140 IF BX<150 THEN PUTSPRITE1,(15,CY),1,1
1150 IF BX>150 THEN PUTSPRITE 2,(225,FY),1,2
1160 COLOR 1,14
1170 SCREEN 1:KEY OFF
1180 LOCATE 1,6:PRINT"HAS DESTRUIDO LA FORTALEZA "
1190 LOCATE 10,8:PRINT"ENEMIGA"
1200 LOCATE 0,11:PRINT"ENHORABUENA JUGADOR NUMERO ";Q$
1210 LOCATE 1,15:PRINT"OTRA PARTIDA SI O NO [S/N]"
1220 A$=INKEY$
1230 IF A$="S" OR A$="s" THEN CLEAR:GOTO 10
1240 IF A$="n" OR A$="N" THEN END
1250 GOTO 1220

```





HT-40


```
10 / *****
20 / *****
30 / ***** HT-40 *****
40 / *****
50 / *****
60 /
70 /EUGENIO GARRIDO GOMEZ
80 /
90 /      1986
100 / *****
110 /
120 COLOR 3,6,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(45,80),6:PRINT#1,"HT-40
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1
200 LOCATE 3,10:PRINT"JOTSTICK O CURSOR(J/C)"
210 X$=INKEY$
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X$="J"OR X$="j" THEN CT=1:GOTO270
250 IF X$="C"OR X$="c" THEN CT=0:GOTO270
260 GOTO 210
270 COLOR 10,4,4:SCREEN2,2
280 GOSUB 630:X1=141:X2=141:X3=141
290 TIME=0
300 ON SPRITE GOSUB 500:SPRITE ON
310 PUTSPRITE 1,(42,X1), 8, 1
320 PUTSPRITE 2,(127,X2), 8, 1
330 PUTSPRITE 3,(212,X3), 8, 1
340 SPRITE ON
350 A=STICK(CT)
360 IF A=7 AND DI=0 THEN DI=1:T1=1
370 IF A=1 AND DI=0 THEN DI=1:T2=1
380 IF A=3 AND DI=0 THEN DI=1:T3=1
390 IF T1=1THEN X1=X1-3
400 IF T2=1THEN X2=X2-3
410 IF T3=1THEN X3=X3-3
420 IF X1=<0THEN X1=141:DI=0:T1=0
430 IF X2=<0THEN X2=141:DI=0:T2=0
440 IF X3=<0THEN X3=141:DI=0:T3=0
```



```

450 PUT SPRITE7,(HX,HY),1,3
460 HX=HX-5:IF HX<=0 THEN HY=INT(RND(1)*125):HX=255
470 IF TIME =6000 THEN 590
480 GOTO 310
490 PUTSPRITE 2,(127,X2), 8, 1
500 SPRITE OFF:IF HX>38ANDHX<46THEN X1=141:DI=0:T1=0:PU=PU+10:GOSUB 580
510 IF HX>123ANDHX<132THEN X2=141:DI=0:T2=0:PU=PU+10:GOSUB 580
520 IF HX>208ANDHX<216THEN X3=141:DI=0:T3=0:PU=PU+10:GOSUB580
530 PUTSPRITE 1,(42,X1), 8, 1
540 PUTSPRITE 2,(127,X2), 8, 1
550 PUTSPRITE 3,(212,X3), 8, 1
560 IF DI=0THEN HY=INT(RND(1)*125):HX=255
570 SPRITE ON:RETURN
580 SOUND 6,&HF:SOUND7,&H7:SOUND8,&H10:SOUND9,&H10:SOUND10,&H10:SOUND12,&H40:SOU
ND 13,&H0:RETURN
590 OPEN"GRP:"AS#1
600 PSET(80,100):PRINT#1,"PUNTOS=";PU
610 FOR N=1 TO2000:NEXT N
620 RUN
630 FOR K=1 TO 3 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$(K)=B$:NEXT
640 LINE (0,170)-(255,192),10,BF
650 PUTSPRITE 4,(42,157), 1, 2
660 PUTSPRITE 5,(127,157), 1, 2
670 PUTSPRITE 6,(212,157), 1, 2
680 RETURN
690 DATA1,1,3,3,3,3,6,A,A,A,E,3,1,3,2,9,0,0,80,80,80,80,C0,A0,A0,A0,E0,80,0,80,2
0,40,0,1,1,1,1,1,1,1,3,3,12,BB,BA,92,93,FF,0,0,0,0,0,0,0,41,41,AA,9C,88,94,B6,FE,E
B,FF,0,0,0,0,0,FF,2,6,F,13,63,FF,7F,22,91,7F,0,0
700 DATA0,0,0,F8,0,C3,7,85,FF,FC,80,0,0,C0

```



SALVAMENTO

```

10 ' *****
20 ' *****
30 ' ***** SALVAMENTO *****
40 ' *****
50 ' *****
60 '
70 ' LUIS SANGUINO ARIAS

```



```

80 /
90 /      1986
100 / *****
110 /
120 COLOR 15,12,1:KEY OFF
130 SCREEN 3
140 OPEN"GRP:"AS1
150 PSET(10,80),12:PRINT#1,"SALVAME!"
160 FOR I=1 TO 1500
170 NEXT
180 CLOSE
190 COLOR 1,4,1
200 SCREEN 2, 2
210 OPEN"grp:"AS1
220 DEFINT X,Y,J,K
230 FOR K=1 TO 4 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$(K)=B$:NEXT
240 AA=1:BA=2:HU=0:HI=3
250 GOSUB 460
260 CIRCLE(227,90),15,15
270 CIRCLE(227,90),10,15
280 CIRCLE(227,90),7,15
290 LINE(0,93)-(255,0),5,BF
300 FOR ES=0 TO 255 STEP 2
310 YE=INT(RND(1)*95)
320 PSET(ES,YE),INT(RND(1)*14)+1
330 NEXT
340 CIRCLE(55,35),15,11
350 PAINT(50,35),11
360 CIRCLE(65,35),15,5
370 PAINT(65,35),5
380 TIME=0
390 ON STRIGGOSUB 430
400 STRIG(CT) ON
410 TI=TIME/50
420 GOTO 420
430 STRIG(CT) OFF
440 X=X+1
450 IF X>210 THEN 620
460 PUTSPRITE 1 ,(X,85), 1,AA
470 PUTSPRITE 2 ,(220,80+HU),1,3
480 IF RR=0 THEN RR=1 :RETURN
490 IF TI>35 THEN 520
500 SWAP AA,BA:SWAP HU,HI
510 GOTO 390
520 YH=85
530 CIRCLE(227,90),15,4
540 CIRCLE(227,90),10,4
550 CIRCLE(227,90),7,4
560 LINE(210,70)-(245,93),5,BF
570 PUTSPRITE 2 ,(220,209),1,3
580 YH=YH+1:IF YH>192 THEN MU=1:GOTO 620
590 PUTSPRITE 3 ,(220,YH),1,4
600 FOR I=1 TO25:NEXT
610 GOTO 580
620 COLOR 1,14
630 SCREEN 1
640 IF MU=1 THEN LOCATE 0,6:PRINT"      LO SIENTO      TU CHICA SE HA
AHOGADO"
650 IF MU=0 THEN LOCATE 0,6:PRINT"      ERES UN HEROE.....ELLA TE AMARA
SIEMPRE"
660 LOCATE 2,10:PRINT"HAS TARDADO ";TI;"SEGUNDOS"
670 LOCATE 2,20:PRINT"OTRA PARTIDA SI O NO [S/N]"


```



```

680 A$=INKEY$
690 I=I+1:IFI>13 THEN I=0
700 VDP(7)=I
710 IF A$="S" OR A$="s" THEN CLEAR:GOTO190
720 IF A$="n" OR A$="N" THEN END
730 GOTO 680
740 DATA 0,0,0,0,3,E7,9F,8,F0,80,0,0,0,0,0,0,8,14,22,ED,FC,CC,20,10,10,10,20,
0,0,0,0,0,0,0,0,C3,27,1F,8,11,26,C0,80,0,0,0,0,0,0,0,C,EC,FE,CD,81,0,0,0,0,0,0
750 DATA 9,5,3,2,C4,25,12,9,7,3,3,3,1,0,0,0,20,40
760 DATA 80,80,46,48,90,20,C0,80,80,80,0,0,0,0,0,D,3,4,5,2,1,47,AB,11,1,3,7,8,5,
2,80,40,80,40,46,88,10,E0,C0,80,80,C0,E0,10,A,4

```



CUESTA

```

10 '*****
20 '****          ****
30 '****  CUESTA  ****
40 '****          ****
50 '*****
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 '          1986
100 '*****
110 '
120 COLOR 1,3,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(40,80),3:PRINT#1,"CUESTA"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1
200 LOCATE 3,10:PRINT"JOTSTICK 0 CURSOR(J/C)"
210 X$=INKEY$

```



```

220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X$="J"OR X$="j" THEN CT=1:GOTO270
250 IF X$="c"OR X$="C" THEN CT=0:GOTO270
260 GOTO 210
270 COLOR 1,15,8:SCREEN 2,2,1
280 OPEN"GRP:"AS#1
290 DRAW"C3BM0,180R29U21E25R19E4U20E12R37E18R15E6U20R16E20U1R16E5U15R11E16U1D1
300 PAINT (1,191),3
310 FOR K=1 TO 7 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRI1
E$ (K)=B$:NEXT
320 VI=5
330 GOSUB 750
340 X=5:Y=164
350 A=1:B=2:PX=240:PY=-4:E=6:G=7
360 ON SPRITE GOSUB 590
370 PX=PX-4:PY=PY+4
380 IF POINT(PX+8,PY+26)<>3 THEN PY=PY+3
390 IF POINT(PX+8,PY+18)=3 THEN PY=PY-3:PX=PX-3
400 IF PX<=0 THEN PY=-4:PX=240
410 IF F=0 THEN PUTSPRITE 1,(X,Y),1,A
420 IF F=1 THEN PUTSPRITE 1,(X,Y),1,C
430 IF F=2 THEN PUTSPRITE 1,(X,Y),1,5
440 PUTSPRITE 2,(PX,PY),6,E
450 GOSUB 490
460 IF MM=0 THEN SPRITE ON
470 SWAP E,G
480 GOTO 370
490 IF MM=0 THEN S=STICK(CT)
500 IF S=3 THEN X=X+2:Y=Y-2:SWAP A,B:F=0
510 IF S=7 THEN X=X-2:Y=Y+2:SWAP C,D:F=1
520 IF S=5 THEN F=2
530 IF POINT(X+12,Y+16)<>3 THEN Y=Y+2
540 IF POINT(X+12,Y+14)=3 THEN Y=Y-2
550 IF Y<=-8 THEN 630
560 IF X<=0 AND MM=1 THEN MM=0 :GOSUB 750
570 IF X<=0 THEN X=0
580 RETURN
590 SPRITE OFF
600 VI=VI-1
610 MM=1:S=7:D=3:C=5:PY=209
620 RETURN
630 PU=PU+100:BEEP:BEEP:BEEP
640 LINE (70,180)-(180,192),15,BF
650 PSET(76,182),15:PRINT#1,"PUNTOS=";PU
660 GOTO 340
670 SCREEN 1:KEYOFF
680 LOCATE 8,10:PRINT"HAS MUERTO"
690 LOCATE 3,15:PRINT"TUS PUNTOS HAN SIDO=";PU
700 LOCATE 1,20:PRINT"OTRA PARTIDA SI O NO [S/N]"
710 X$=INKEY$
720 IF X$="S"OR X$="s"THEN CLEAR:GOTO 270
730 IF X$="n"OR X$="N"THEN SCREEN 0:END
740 GOTO 710
750 LINE(80,0)-(155,10),13,BF
760 PSET(86,2),3:PRINT#1,"VIDAS=";VI
770 IF VI=0 THEN 670
780 C=3:D=4:F=2
790 IF BB=0 THEN BB=1:F=0
800 RETURN
810 DATA 1,3,3,1,0,1,1,1,3,5,1,2,4,38,20,0,80,C0,E0,C0,80,80,80,F0,80,80,80,40,
0,10,10,18
820 DATA 1,3,3,1,0,1,1,1,1,3,5,1,1,1,1,1,80,C0,E0,C0,80,80,80,80,C0,A0,90,40,40


```



```

40,40,80
830 DATA 1,3,7,3,1,1,1,F,1,1,1,2,4,8,8,1B,80,C0,C0,80,0,80,80,80,C0,A0,80,40,2C
1C,4,0
840 DATA 1,3,7,3,1,1,1,1,3,5,9,2,2,2,2,1,80,C0,C0,80,0,80,80,80,80,C0,A0,80,80,
0,80,80
850 DATA 0,0,0,0,0,0,0,0,0,0,0,E3,9F,7,EB,90,0,0,0,0,0,0,0,0,0,C,CE,FF,CE,44,3
860 DATA 0,1,7,F,F,1B,37,37,6F,6F,37,1B,F,F,1,0,0,80,80,C0,C0,E0,F0,70,B8,DB,F8,
0,F0,E0,80,0
870 DATA 0,1,F,F,1B,37,6F,6F,37,37,1B,F,F,7,1,0,0,80,E0,F0,F0,F8,DB,B8,70,F0,E0
C0,C0,80,80,0,0

```



XR-2000

```

10 '*****
20 '*****
30 '***** XR-2000 *****
40 '*****
50 '*****
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 ' 1986
100 '*****
110 '
120 COLOR 15,12,1
130 SCREEN 3
140 OPEN"GRP:"AS1
150 PSET(25,80),12:PRINT#1,"XR-2000"
160 FOR I=1 TO 1500
170 NEXT
180 CLOSE
190 COLOR 1,7,8
200 SCREEN 2, 2 :FOR K=1 TO 1 :B$="":FOR I=1 TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+f
)):NEXT:SPRITE$(K)=B$:NEXT
210 OPEN"grp:"AS1
220 CIRCLE(50,35),15,10


```




```

230 PAINT(50,35),10
240 LINE (0,80)-(255,170),12,BF
250 LINE (0,170)-(255,192),4,BF
260 PSET(75,70),7:PRINT#1,"!! LISTOS !! "
270 FOR I=1 TO 700:NEXT
280 LINE (75,70)-(225,78),7,BF
290 SOUND 6,15:SOUND 7,7:SOUND 8,16:SOUND 9,16:SOUND 10,16:SOUND 11,0 :SOUND 1:
16:SOUND 13,0
300 K=127
310 G=289
320 ON STRIGGOSUB 570
330 STRIG(CT) ON
340 LINE (J,177)-(J+10,181),10,BF
350 LINE (K,177)-(K+10,181),10,BF
360 IF X<0 THEN X=0
370 R=R+.52
380 T=T+.47
390 V=V+.38
400 IF X>230 THEN 590
410 IF R>230 THEN PR=1 :GOTO 590
420 PUTSPRITE 1 ,(X,165), 1, 1
430 PUTSPRITE 2 ,(T,174), 3, 1
440 PUTSPRITE 3 ,(V,167), 11,1
450 PUTSPRITE 4 ,(R,160), 15, 1
460 IF B=1 THEN 550
470 LINE (J+10,177)-(J+20,181),4,BF
480 LINE (K+10,177)-(K+20,181),4,BF
490 IF G>255 THEN 520
500 LINE(G,100)-(G+2,170),15,BF
510 LINE(G+2,110)-(G+4,170),12,BF
520 G=G-.4:IF G<=215 THEN G=215 :GOTO 560
530 K=K-3:IF K<-10 THEN K=255
540 J=J-3:IF J<-10 THEN J=255
550 GOTO 330
560 PSET(G+7,103),1:PRINT#1,"META" :LINE(238,170)-(242,192),14,BF:B=1:GOTO 550
570 STRIG(CT) OFF:X=X+1:RETURN
580 DATA 0,0,0,1,2,5,B,E,F,5,39,49,9F,85,48,30,10,38,30,C0,40,C0,34,1C,8,8,7C,9,
29,A5,12,C
590 COLOR 1,14
600 SCREEN 1
610 IF PR=1 THEN 640
620 LOCATE 4,7:PRINT"ENHORABUENA CAMPEON"
630 IF PR=0 THEN 650
640 LOCATE 4,6:PRINT"LO SIENTO HAS PERDIDO"
650 LOCATE 2,11:PRINT"OTRA PARTIDA SI O NO [S/N]"
660 A$=INKEY$
670 I=I+1:IFI>13 THEN I=0
680 VDP(7)=I
690 IF A$="S" OR A$="s" THEN CLEAR:GOTO 190
700 IF A$="n" OR A$="N" THEN END
710 GOTO 660

```





ATERRIZAJE

```
10 / *****
20 / ****          ****
30 / ****ATERRIZAJE ****
40 / ****          ****
50 / *****
60 /
70 /EUGENIO GARRIDO GOMEZ
80 /
90 /      1986
100 / *****
110 /
120 COLOR 3,6,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(10,80),6:PRINT#1,"ATERRIZA"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1
200 LOCATE 3,10:PRINT"JOTSTICK O CURSOR(J/C)"
210 X#=INKEY#
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X#="J"OR X#="j" THEN CT=1:GOTO270
250 IF X#="C"OR X#="c" THEN CT=0:GOTO270
260 GOTO 210
270 CLS:KEY OFF:COLOR 15,1,1:SCREEN 2,2:OPEN "GRP:"AS #1:
280 X=1
290 H=RND(-TIME)
300 FOR W=1 TO 260
310 X=X+1:IF X>15 THEN X=2
320 K=INT(RND(1)*192)
330 PSET(W,K),X
340 NEXT
350 GOSUB 550:ON SPRITE GOSUB 520:SPRITE ON:MY=209:CO=100:M2=1
360 LINE(0,180)-(CO+1,184),1,BF:LINE(0,180)-(CO,184),6,BF
370 PUTSPRITE 1,(NX,NY), 10, 1
380 PUTSPRITE 3,( AX+16,150), 2, 2
390 PUTSPRITE 2,( AX,150), 2, 2
400 PUTSPRITE 4,(MX,MY), 6, 3
410 AX=AX+3:IF AX=255 THEN AX=0
420 A=STICK(CT)
430 IF A=7THEN M1=1:M2=0
```



```

440 IF A=3 THEN M2=1:M1=0
450 IF M1=1 THEN IF NX>0 THEN NX=NX-3:M2=0
460 IF M2=1 THEN IF NX<255 THEN NX=NX+3:M1=0
470 NY=NY+3:IF NY>138 THEN NY=0:GOSUB 570
480 B=STRIG(CT)
490 IFB=-1 THEN MX=NX:MY=NY+12:NY=NY-4:CO=CO-1:LINE(0,180)-(CO+1,184),1,BF:LINE
,180)-(CO,184),6,BF:ELSE MY=209
500 IF CO=<0 THEN 540
510 SPRITE ON:GOTO 370
520 BEEP:IF NX>AX+3 AND NX<AX+27 AND NY>134 THEN NA=NA+1:PLAY"ABABBACAD":FOR N=
TO 2000:NEXT N:CO=CO+10:PU=PU+1
530 SPRITE OFF:RETURN
540 PSET (50,100):PRINT#1,"ATERRIZASTE ";PU;" VECES":FOR N=1 TO 4000:NEXT:RUN
550 FOR K=1 TO 3 :B$="":FOR I=1 TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPR
E$(K)=B$:NEXT
560 RETURN
570 SOUND 6,&HF:SOUND 7,&H7:SOUND 8,&H10:SOUND 9,&H10:SOUND 10,&H10:SOUND 12,&H40:SO
ND 13,&H0
580 FOR N=1 TO 10:PUTSPRITE 1,(NX,NY),N,1:FOR H=1 TO 20:NEXT H:NEXT N
590 RETURN
600 DATA 7,9,13,2F,27,13,9,F,D,9,F,9,12,20,70,0,E0,90,CB,F4,E4,CB,90,F0,B0,90,F0
90,48,4,E,0,FF,FF,7F,7E,5E,4A,2,2,0,0,0,0,0,0,0,0,FF,FF,FF,F5,D4,90,10,0,0,0,0
,0,0,0,0,1,0,2,4,2,5,2,E,4,1,0,0,0,0,0,0,80,A0
610 DATA 0,C0,90,50,A0,90,60,80,0,0,0,0,0,0


```

**SI PREFIERE SUSCRIBIRSE
POR TELÉFONO...**

LLAME AL

(91) 266 66 01
(91) 266 66 01





BOMBEROS

```
10 '*****
20 '****          ****
30 '**** BOMBEROS ****
40 '****          ****
50 '*****
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 '          1986
100 '*****
110
120 COLOR 1,10,4
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(5,80),10:PRINT#1,"BOMBEROS"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 12,14,6
190 SCREEN 2,2
200 OPEN"grp:"AS1
210 DIM X(3),Y(3)
220 TIME=0
230 FOR F=1 TO 3:READ X(F):NEXT
240 FOR F=1 TO 3:READ Y(F):NEXT
250 YM=209:X=100
260 DATA 168,58,118,4,7,27
270 FOR K=1 TO 4 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPR:
E$ (K)=B$:NEXT
280 LINE(0,170)-(255,192),4,BF
290 LINE(0,175)-(80,183),1,BF
300 PSET(4,176),1:PRINT#1,"VIDAS=";VI
310 LINE(0,172)-(255,167),14,BF
320 LINE(0,167)-(255,110),15,BF
330 LINE(40,130)-(20,0),12,BF
340 LINE(20,150)-(40,130),12
350 LINE(20,150)-(20,130),12
360 PAINT(22,140),12
370 LINE(20,150)-(0,0),3,BF
380 LINE(7,150)-(0,130),1,BF
390 LINE(41,146)-(200,0),7,BF
400 LINE(241,110)-(200,0),5,BF
410 LINE(241,110)-(200,146),5
```



```

420 LINE(200,110)-(200,146),5
430 PAINT(202,115),5
440 LINE(170,146)-(190,106),8,BF
450 CIRCLE(242,153),7,2,,,5
460 PAINT(242,155),2
470 FOR I=1 TO 255 STEP 30
480 LINE(I,189)-(I+13,192),11,BF
490 NEXT
500 PSET(173,96),7:PRINT#1,"43"
510 INTERVAL ON
520 ON INTERVAL=70 GOSUB 730
530 D=STICK(CT)
540 U=U+1:IF U>11 THEN U=6
550 VDP(7)=U
560 IF TIME>4000 THEN 980
570 ON SPRITE GOSUB 810
580 LINE(120,40)-(130,25),11,BF
590 LINE(70,20)-(60,5),11,BF
600 LINE(170,17)-(180,2),11,BF
610 IF D=3 THEN X=X+3
620 IF D=7 THEN X=X-3
630 PUTSPRITE 3 ,(XM,YM), 1, 3
640 PUTSPRITE 1 ,(X,145), 1, 1
650 PUTSPRITE 2 ,( X+19,145), 1, 2
660 IF CA=1 THEN YC=YC+9:PUTSPRITE 4 ,( XC, YC), 1, 4
670 IF CA=1 AND YC>146 THEN GOSUB880
680 SPRITE ON
690 LINE(120,40)-(130,25),6,BF
700 LINE(70,20)-(60,5),6,BF
710 LINE(170,17)-(180,2),6,BF
720 GOTO 530
730 YC=YM:XC=XM:YM=209
740 IF T=0 THEN 770
750 INTERVAL OFF
760 CA=1
770 T=1
780 CO=INT(RND(1)*3)+1
790 XM=X(CO):YM=Y(CO)
800 RETURN
810 IF YC>100 THEN VI=VI+1 ELSE 840
820 LINE(50,175)-(80,183),1,BF
830 PSET(50,176),1:PRINT#1,VI
840 YC=209:CA=0
850 PUTSPRITE 4 ,( XC, YC), 1, 4
860 INTERVAL ON
870 SPRITE OFF:RETURN
880 PLAY"t230m200o1dcC"
890 CIRCLE(XC+8,YC+7),8,6
900 PAINT(XC+8,YC+8),6
910 FOR I=1 TO 25:NEXT
920 CIRCLE(XC+8,YC+7),8,15
930 PAINT(XC+8,YC+8),15
940 CA=0
950 YC=209:INTERVAL ON:RETURN
960 DATA60,F8,D0,88,50,20,78,74,7B,27,70,70,50,50,50,68,0,0,0,0,0,0,1F,F0,10,10
F0,1F,0,0,0,0,0,0,0,0,0,0,0,0,F8,F,8,8,F,F8,0,0,0,0,0,6,1F,B,11,A,4,1E,2E,DE,E4,E,E,4
A,A,16,9,5,3,2,C4,25,12,9,7,3,3,3,1,0,0,0,20,40
970 DATA80,80,46,48,90,20,C0,80,80,80,0,0,0,0,0,D,3,4,5,2,1,47,AB,11,1,3,7,8,5
,80,40,80,40,46,88,10,E0,C0,80,80,C0,E0,10,A,4
980 COLOR 1,14
990 SCREEN 1:KEY OFF
1000 LOCATE 3,4:PRINT "TU MISION HA TERMINADO"
1010 LOCATE 1,8:PRINT "HAS LOGRADO SALVAR";VI;"VIDAS"


```



```

1020 LOCATE 1,14:PRINT "OTRA PARTIDA SI O NO [S/N]"
1030 X$=INKEY$
1040 IF X$="S" OR X$="s" THEN CLEAR:GOTO 180
1050 IF X$="n" OR X$="N" THEN END
1060 GOTO 1030

```



MEMORY-TEST

```

10 '*****
20 '*****
30 '*****MEMORY-TEST*****
40 '*****
50 '*****
60 '
70 'EUGENIO GARRIDO GOMEZ
80 '
90 '      1986
100 '*****
110 '
120 COLOR 3,6,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(35,30),6:PRINT#1,"MEMORY"
160 PSET(60,90),6:PRINT#1,"TEST"
170 FOR Y=1 TO 1500:NEXT
180 CLOSE
190 COLOR 15,12,12:SCREEN 2
200 A1=1:B1=1:C1=1:D1=1:E1=1
210 OPEN "GRP:"AS#1
220 PSET(80,10),10:PRINT#1,"MEMORY-TEST"
230 PSET(81,10),10:PRINT#1,"MEMORY-TEST"
240 PSET(45,30),10:PRINT#1,"COMPRUEBA TU MEMORIA"
250 PSET(46,30),10:PRINT#1,"COMPRUEBA TU MEMORIA"
260 DEFFN X=INT(RND(-TIME)*8)+1

```



```

270 FORN=1 TO10
280 PLAY"AB"
290 A=FN X:B=FNX:C=FNX:D=FNX:E=FNX
300 GOSUB310:NEXTN:GOTO 470
310 CIRCLE (40,100),10,A
320 PAINT (40,100),A
330 PSET (30,98),A:PRINT#1,A
340 CIRCLE (80,100),10,B
350 PAINT (80,100),B
360 PSET (70,98),B:PRINT#1,B
370 CIRCLE (120,100),10,C
380 PAINT (120,100),C
390 PSET (110,98),C:PRINT#1,C
400 CIRCLE (160,100),10,D
410 PAINT (160,100),D
420 PSET (150,98),D:PRINT#1,D
430 CIRCLE (200,100),10,E
440 PAINT (200,100),E
450 PSET (190,98),E:PRINT#1,E
460 RETURN
470 NF$=STR$(A)+STR$(B)+STR$(C)+STR$(D)+STR$(E)
480 FOR N=0 TO 20:BEEP:NEXTN
490 GOSUB 770:GOSUB 830
500 PSET (20,150):PRINT#1,"INTRODUCE CODIGO DE COLORES"
510 PSET (21,150):PRINT#1,"INTRODUCE CODIGO DE COLORES"
520 LINE (48,178)-(95,190),6,BF
530 NP$="":FOR N=0 TO 4
540 F$=INKEY$
550 IF F$="" THEN 540
560 IF ASC(F$)=8 THEN 500
570 IF F$ <"0"ORF$>"9" THEN 540
580 PSET (50+(N*8),180),6:PRINT#1,F$
590 NP$=NP$+" "+F$
600 NEXT N
610 IF NF$=NP$ THEN 730
620 GOSUB 770:GOSUB 310
630 LINE (20,148)-(236,158),12,BF
640 PSET (20,150):PRINT#1,"TU MEMORIA HA FALLADO"
650 PSET (21,150):PRINT#1,"TU MEMORIA HA FALLADO"
660 FOR N=1 TO 1000:NEXT
670 LINE (0,148)-(255,158),12,BF
680 PSET (0,150):PRINT#1,"QUIERES VOLVER A INTENTARLO(S/N)"
690 PSET (1,150):PRINT#1,"QUIERES VOLVER A INTENTARLO(S/N)"
700 IF INKEY$="S" THEN RUN 190
710 IF INKEY$="N" THEN STOP
720 GOTO 700
730 LINE (0,148)-(255,158),12,BF
740 PSET (0,150):PRINT#1,"ENHORA BUENA ACERTASTE":PLAY "ABCABCCD"
750 PSET (1,150):PRINT#1,"ENHORA BUENA ACERTASTE"
760 GOTO 660
770 SWAP A,A1
780 SWAP B,B1
790 SWAP C,C1
800 SWAP D,D1
810 SWAP E,E1
820 RETURN
830 CIRCLE (40,100),10,A
840 PAINT (40,100),A
850 CIRCLE (80,100),10,B
860 PAINT (80,100),B
870 CIRCLE (120,100),10,C
880 PAINT (120,100),C
890 CIRCLE (160,100),10,D


```



```

900 PAINT (160,100),D
910 CIRCLE (200,100),10,E
920 PAINT (200,100),E
930 RETURN

```



PROGRAMA PISTA

```

10 / ***** PROGRAMA *****
20 / ***** *****
30 / ***** PISTA *****
40 /
50 / ** LUIS SANGUINO 1986 **
60 /
70 COLOR 15,5,1:KEY OFF
80 SCREEN 3,3
90 OPEN "grp:"AS #1
100 PRESET (50,5),1:PRINT#1,"PISTA"
110 FOR I=0 TO 31:READ A
120 VPOKE(14336+I),A
130 NEXT
140 DATA 0,31,0,128,195,199,253,118,29,7,3,1,33,31,0,0,0,255,64,64,24
,97,254,252,32,33,254,0,0
150 FOR I=0 TO 13:READ S:SOUND I,S:NEXT
160 DATA 5,5,0,1,24,0,9,3,2,2,16,90,2,12
170 VPOKE 6912,100
180 VPOKE 6913,100
190 VPOKE 6914,0
200 VPOKE 6915,1
210 LINE(0,160)-(255,192),10,BF
220 FOR I=1 TO 2500:NEXT
230 CLOSE
240 SCREEN 1


```




```

250 COLOR 1,7,6
260 LOCATE 1,6:PRINT"PULSA 0 PARA TECLADO"
270 LOCATE 1,8:PRINT"      1 PARA JOYSTICK-1"
280 LOCATE 1,10:PRINT"      2 PARA JOYSTICK-2";
290 INPUT JT
300 IF JT>2 OR JT<0 THEN 290
310 COLOR 15,1,1
320 SCREEN 2,2,1
330 RESTORE 140
340 FOR I=0 TO 31:READ A
350 VPOKE(14336+I),A :NEXT
360 OPEN "grp:"AS #1
370 Z=RND(-TIME)
380 DIM A(40)
390 N=RND(-TIME)
400 FOR EX=1 TO 255
410 EY=20+INT(RND(1)*170)
420 PSET(EX,EY),INT(RND(1)*15)
430 NEXT
440 CIRCLE(30,40),10,14
450 PAINT(30,40),14
460 CIRCLE(35,40),10,1
470 PAINT(35,40),1
480 LINE(60,0)-(190,12),5,BF
490 PSET(66,2),5:PRINT#1,"ATERRIZAJES=";AT
500 FOR I=1 TO 500:NEXT
510 LINE (0,180)-(255,192),12,BF
520 XP=INT(RND(1)*240)
530 LINE(XP,180)-(XP+25,185),11,BF
540 XH=0:YH=20
550 XH=XH+8
560 XH=XHMOD(256)
570 PUT SPRITE 0,(XH,YH),15,0
580 IF POINT(XH+8,YH+15)=11 THEN AT=AT+1:GOTO 480
590 GOSUB 620
600 IF POINT(XH+2,YH+15)=12 THEN 660
610 GOTO 550
620 D=STICK(JT)
630 IF D=5 THEN YH=YH+2
640 IF D=1 THEN YH=YH-2
650 RETURN
660 SCREEN 1
670 LOCATE 6,7:PRINT "      LO SIENTO"
680 LOCATE 5,10:PRINT "TE HAS ESTRELLADO"
690 LOCATE 0,15:PRINT "HAS CONSEGUIDO";AT;"ATERRIZAJES"
700 LOCATE 1,20:PRINT "LO INTENTAS DE NUEVO [S/N]"
710 A$=INKEY$
720 IF A$="S" OR A$="s" THEN CLEAR :GOTO 10
730 IF A$="n" OR A$="N" THEN CLS:END
740 GOTO 710

```





ATAQUE

```
10 '*****
20 '****          ****
30 '****    ATAQUE    ****
40 '****          ****
50 '*****
60 '
70 'LUIS SANGUINO ARIAS
80 '
90 '          1986
100 '*****
110 '
120 COLOR 2,14,1
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(40,80),14:PRINT#1,"ATAQUE"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1
200 LOCATE 3,10:PRINT"JOTSTICK O CURSOR(J/C)"
210 X#=INKEY$
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X#="J"OR X#="j" THEN CT=1:GOTO270
250 IF X#="c"OR X#="C" THEN CT=0:GOTO270
260 GOTO 210
270 COLOR 1,1,1
280 SCREEN 2, 2 :FOR K=1 TO 6 :B#="" :FORI=1TO 32 :READ A#:B#=B#+CHR$(VAL("&H"+A#
)):NEXT:SPRITE$(K)=B#:NEXT
290 LINE(0,18)-(255,0),10,BF
300 OPEN "grp:"AS #1
310 EN=1500
320 PSET(5,5),10:PRINT#1,"ENERGIA=";EN
330 PSET(155,5),10:PRINT#1,"PUNTOS=";PT
340 Z=RND(-TIME)
350 LINE(0,19)-(255,192),1,BF
360 N=RND(-TIME)
370 FOR EX=1 TO 255 STEP 2
380 FF=(FF+1)MOD(15)
390 EY=20+INT(RND(1)*170)
```




```

400 PSET (EX,EY),FF
410 NEXT
420 X=120:Y=100:A=1:X1=128:Y1=30:Z=1:Z1=1
430 PUTSPRITE 5,(55,170),10,5
440 PUTSPRITE 6,(125,170),10,5
450 PUTSPRITE 7,(200,170),10,5
460 INTERVAL ON:ON INTERVAL=20 GOSUB 720
470 IF Y1>60 THEN A=2 :IF Y1>100 THEN A=3
480 Y1=Y1+Z:IF Y1>150 THEN Z=-1
490 IF Y1>140 THEN GOSUB 590
500 X1=(X1+Z1)MOD(255)
510 PUTSPRITE 3,(X1,Y1),3,A
520 PUTSPRITE 4,(X,Y),15,4
530 IF EN<0 THEN 900
540 GOSUB 770
550 IF STRIG(CT)=-1 THEN GOSUB 690
560 GOTO 470
570 '
580 '
590 LINE(X1+8,Y1+8)-(X1+8,180),13:BEEP
600 IF Z=0 THEN 620
610 INTERVAL OFF:Z=0:IF X1>128 THEN Z1=-1 ELSE Z1=1
620 IF R1=0 THEN IF X1+8>55 AND X1+8<70 THEN PUTSPRITE 5,(55,170),6,6:EN=EN-200:
R1=1
630 IF R2=0 THEN IF X1+8>125 AND X1+8<140 THEN PUTSPRITE 6,(125,170),6,6:EN=EN-2
00:R2=1
640 IF R3=0 THEN IF X1+8>200 AND X1+8<216 THEN PUTSPRITE 7,(200,170),6,6:EN=EN-2
00:R3=1
650 LINE(X1+8,Y1+8)-(X1+8,180),1
660 EN=EN-2
670 GOSUB 870
680 RETURN
690 IF X+8>X1 AND X+8<X1+14 AND Y+8>Y1 AND Y+8<Y1+12 THEN A=1:Y1=30:X1=128:INTER
VAL ON:PT =PT+50:EN=EN+20:GOSUB 870
700 EN=EN-20
710 RETURN
720 F=INT(RND(1)*4)
730 G=INT(RND(1)*2)
740 IF G=0 THEN Z1=2.5 ELSE Z1=-2.5
750 IF F=0 THEN Z=-2.5 ELSE Z=2.5
760 RETURN
770 D=STICK(CT)
780 IF D=1 THEN Y=Y-1.5
790 IF D=2 THEN Y=Y-1:X=X+1
800 IF D=3 THEN X=X+1.5
810 IF D=4 THEN Y=Y+1:X=X+1
820 IF D=5 THEN Y=Y+1.5
830 IF D=6 THEN Y=Y+1:X=X-1
840 IF D=7 THEN X=X-1.5
850 IF D=8 THEN Y=Y-1:X=X-1
860 RETURN
870 LINE(70,5)-(110,13),10,BF:PSET(70,5),10:PRINT#1,EN
880 LINE(210,5)-(250,13),10,BF:PSET(210,5),10:PRINT#1,PT
890 RETURN
900 COLOR 1,15,2:SCREEN 1:KEY OFF
910 LOCATE 4,10:PRINT"LO SIENTO TU RESERVA"
920 LOCATE 2,12:PRINT"DE ENERGIA SE HA AGOTADO"
930 LOCATE 3,16:PRINT"PUNTOS CONSEGUIDOS=";PT
940 DATA0,0,0,0,0,0,3,5,5,2,6,C,0,0,0,0,0,0,0,0,0,0,C0,A0,A0,40,60,30,0,0,0,0,0,
0,0,0,8,7,9,9,9,6,4,8,3C,0,0,0,0,0,0,0,10,E0,90,90,90,60,20,10,3C,0,0,0,10,B,7,1
9,31,35,31,19,7,6,C,18,30,30,30,F8,8,D0
950 DATAE0,98,8C,AC,8C,98,E0,60,30,18,C,C,C,1F,0,0,0,3,4,8,10,10,1F,10,10,8,4,3,
0,0,0,0,0,E0,90,88,84,84,FC,84,84,88,90,E0,0,0,1,3,3,3F,24,13,15,9,4,2,3,5,9,9,1

```



```
2,3E,80,C0,C0,FC,24,C8,AB,90,20,40,C0,A0,90,90,48,7C
960 DATA 0,0,0,0,0,0,0,0,0,0,0,0,12,37,6D,56,BF,0,0,0,0,0,0,0,0,0,0,20,AB,56,7A,D7
,BD
```



CAÑON


```
10 /*****
20 /****      ****
30 /****   CAÑON   ****
40 /****      ****
50 /*****
60 /
70 /LUIS SANGUINO ARIAS
80 /
90 /      1986
100 /*****
110 /
120 COLOR 11,1,2
130 SCREEN 2,3
140 OPEN"GRP:"AS#1
150 GOSUB 320
160 G=1
170 PUTSPRITE 1 ,(100,110), 6, 1
180 PUTSPRITE 2,(100,20),15,2
190 PSET(25,180),1:PRINT#1,"LUIS SANGUINO ARIAS - 1986"
200 FOR I=1 TO 1600:NEXT
210 CLOSE
220 RESTORE
230 COLOR 1,14,1
240 SCREEN 1
250 LOCATE 2,10:PRINT"JOYSTICK O CURSOR [J/C]"
260 X$=INKEY$
```



```

270 IF X$="J"OR X$="j" THEN CT=1:GOTO 300
280 IF X$="c"OR X$="C" THEN CT=0:GOTO 300
290 GOTO 260
300 COLOR 15,5,1
310 SCREEN 2, 2
320 FOR K=1 TO 3 :B$="":FOR I=1 TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$(K)=B$:NEXT
330 IF G=0 THEN RETURN
340 OPEN"GRP:"AS1
350 X=125:Y=173:RX=0:RY=-20:LX=250:LY=-20:CX=125:CY=-20:BX=133:BY=200
360 LINE (0,170)-(255,192),11,BF
370 LINE (0,180)-(70,192),1,BF
380 PSET (1,182),1:PRINT#1,"PTS=";PT
390 D=STICK(CT)
400 IF D=3 THEN X=X+6
410 IF D=7 THEN X=X-6
420 IF X>240 THEN X=240
430 IF X<0 THEN X=0
440 IF BY<-5 THEN FG=0
450 IF FG=1 THEN 470
460 IF STRIG(CT)=-1 THEN FG=1:BX=X+6:BY=Y-5
470 RX=RX+3:RY=RY+6:LX=LX-3:LY=LY+6:CX=CX+5:CY=CY+6
480 IF FG=1 THEN BY=BY-9
490 SPRITE ON
500 IF CX>255 THEN CX=0
510 IF RX>255 THEN RX=0
520 IF LX<-15 THEN LX=255
530 IF CY>200 THEN CY=0
540 IF RY>200 THEN RY=0
550 IF LY>200 THEN LY=0
560 PUTSPRITE 1 ,(X,Y), 6, 1
570 PUTSPRITE 2,(RX,RY),15,2
580 PUTSPRITE 3,(LX,LY),4,2
590 PUTSPRITE 4,(CX,CY),1,2
600 PUTSPRITE 5,(BX,BY),1,3
610 ON SPRITE GOSUB 640:SPRITE OFF
620 GOTO 390
630 DATA 1,1,1,3,7,7,4,34,36,35,3C,34,34,7,1,C0,40,40,40,60,70,70,10,96,B6,D6,9
E,96,96,F0,C0,C,7,81,41,62,32,3A,1F,1F,F,3,3,3,1,1,1,60,C0,2,4,8C,98,B8,F0,F0,E0
,80,80,80,0,0,0,10,38,7C,38,10,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
,0,0
640 SPRITE OFF
650 IF BX+5>CX AND BX<CX+15 AND BY<CY+16 AND FG=1 THEN CY=0:FG=0:BY=200:GOTO 720
660 IF BX+5>LX AND BX<LX+15 AND BY<LY+16 AND FG=1 THEN LY=0:FG=0:BY=200:GOTO 720
670 IF BX+5>RX AND BX<RX+15 AND BY<RY+16 AND FG=1 THEN RY=0:FG=0:BY=200:GOTO 720
680 IF CX+15>X AND CX<X+15 AND CY+16>180 THEN GOTO 770
690 IF RX+15>X AND RX<X+15 AND RY+16>180 THEN GOTO 770
700 IF LX+15>X AND LX<X+15 AND LY+16>180 THEN GOTO 770
710 GOTO 760
720 PT=PT+50
730 LINE (30,180)-(70,192),1,BF
740 PSET (26,182),1:PRINT#1,PT
750 SOUND 6,15:SOUND 7,7:SOUND 8,16:SOUND 9,16:SOUND 10,16:SOUND 11,0 :SOUND 12,
16:SOUND 13,0
760 RETURN
770 COLOR 1,10,1:SCREEN1
780 LOCATE 0,10:PRINT"LO SIENTO HAS SIDO DESTRUIDO"
790 LOCATE 1,13:PRINT"LO INTENTAS DE NUEVO [S/N]"
800 X$=INKEY$
810 IF X$="S" OR X$="s" THEN CLEAR:G=1:GOTO 230
820 IF X$="n" OR X$="N" THEN END
830 GOTO 800

```

PANZER


```
10 /*****
20 /****          ****
30 /**** PANZER   ****
40 /****          ****
50 /*****
60 /
70 /EUGENIO GARRIDO GOMEZ
80 /
90 /      1986
100 /*****
110 /
120 COLOR 3,6,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(5,80),6:PRINT#1," PANZER"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1:KEY OFF
200 LOCATE 3,10:PRINT"JOTSTICK O CURSOR(J/C)"
210 X#=INKEY#
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X#="J"OR X#="j" THEN CT=1:GOTO270
250 IF X#="C"OR X#="c" THEN CT=0:GOTO270
260 GOTO 210
270 COLOR 1,10,1
280 OPEN"GRP:"AS#1
290 SCREEN 2, 2 :FOR K=1 TO 5 :B#="":FORI=1TO 32 :READ A#:B#=B#+CHR$(VAL("&H"+A#
)) :NEXT:SPRITE$(K)=B#:NEXT
300 C#="":C#=CHR$(240)
310 SPRITE$(6)=C#
320 Y=90:R=2:XE=256:Y2=10
330 ON SPRITE GOSUB 680
340 /
350 /
360 /
370 Y2=Y2+R:E=E+1
380 IF E>=K THEN E=0:K=INT(RND(1)*50):R=-R
```



```

390 IF Y2<=0 THEN :Y2=Y2+4:R=-R
400 IF Y2>=185 THEN Y2=Y2-4:R=-R
410 IF Y<0 THEN Y=0
420 PUTSPRITE 1 ,(17,Y), 1, 1
430 PUTSPRITE 2 ,(220,Y2), 4, 2
440 PUTSPRITE 3 ,(50,42),12, 3
450 GOSUB 560
460 PUTSPRITE 4 ,( 50,132),12, 4
470 PUTSPRITE 5 ,(150,92),12, 4
480 PUTSPRITE 6,(183,160),12, 4
490 PUTSPRITE 7,(80,10),12, 4
500 PUT SPRITE8,(XB,YB),6,3
510 PUT SPRITE9,(XE,YE),6,5
520 GOTO 370
530 '
540 '
550 '
560 D=STICK(CT)
570 IF D=1 THEN Y=(Y-2)MOD 192
580 IF D=5 THEN Y=(Y+2)MOD 192
590 IF Y2>=Y-2 AND Y2<Y+2 AND DE=0 THEN GOSUB 760
600 IF XB>256 THEN GOSUB 700
610 IF XE<0 THEN GOSUB 700
620 IF DE=1 THEN XE=XE-10
630 IF DI=1 THEN XB=XB+9:GOTO 650
640 IF STRIG(CT)=-1 THEN BEEP:SPRITE ON:GOSUB 660
650 RETURN
660 YB=Y+3:XB=37:DI=1
670 RETURN
680 IF XB>200 THEN PU=PU+50
690 IF XE<40 THEN 790
700 IF DI=1 THEN YB=209:DI=0:XB=-10
710 IF DE=1 THEN YE=209:DE=0:XE=256
720 LINE(75,180)-(200,192),10,BF
730 SPRITE OFF
740 PSET (80,180),10:PRINT#1,"PUNTOS=";PU
750 RETURN
760 DE=1:XE=210:YE=Y2+3
770 SPRITE ON
780 RETURN
790 COLOR 15,13,1:SCREEN 1
800 LOCATE 9,10:PRINT"R.I.P"
810 LOCATE 6,15:PRINT"PUNTOS=";PU
820 LOCATE 4,20:PRINT"OTRA PARTIDA S/N"
830 X$= INKEY$
840 IF X$="S" OR X$="s" THEN CLEAR:GOTO 10
850 IF X$="n" OR X$="N" THEN END
860 GOTO 830
40000 DATA 0,0,0,20,10,B,1E,3F,1F,4,3F,6A,9F,92,72,3F,0,0,0,0,0,80,E0,FF,E0,80,FE
,A9,F9,4E,4C,F8
40010 DATA 0,0,0,0,0,1,7,FF,7,1,7F,95,9F,72,32,1F,0,0,0,4,8,D0,78,FC,F8,20,FC,56
,F9,49,4E,FC
40020 DATA 0,0,0,40,10,47,2F,47,90,48,0,0,0,0,0,0,0,0,0,0,80,C0,80,0,0,0,0,0,
0,0,FF,FF,E0,D0,CF,CC,CA,C9,C9,CA,CC,CF,D0,E0,FF,FF,FF,FF,7,B,F3,33,53,93,93,53,
33,F3,B,7,FF,FF
40030 DATA 0,0,0,0,0,0,1,3,1,0,0,0,0,0,0,0,0,0,0,0,2,8,E2,F4,E2,9,12,0,0,0,0,0

```

SARATOGA


```
10 '*****
20 '*****
30 '***** SARATOGA *****
40 '*****
50 '*****
60 '
70 'EUGENIO GARRIDO GOMEZ
80 '
90 '      1986
100 '*****
110 '
120 COLOR 3,6,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(10,80),6:PRINT#1,"SARATOGA"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,14
190 SCREEN 1
200 LOCATE 3,10:PRINT"JOTSTICK O CURSOR(J/C)"
210 X$=INKEY$
220 I=I+1:IFI>13 THEN I=1
230 VDP(7)=I
240 IF X$="J"OR X$="j" THEN CT=1:GOTO270
250 IF X$="C"OR X$="c" THEN CT=0:GOTO270
260 GOTO 210
270 COLOR 15,7,1:SCREEN 2,2:
280 GOSUB 540:PX=117:PY=125
290 PO=PX+16:PP=PX+32:BY=209
300 ON SPRITE GOSUB 410:SPRITE ON
310 PUTSPRITE 2,(PP,PY), 1, 2:PUTSPRITE 1,(PO,PY),1,1:PUTSPRITE 0,(PX,PY),1,0
320 PUT SPRITE3,(AX,0),1,3
330 PUT SPRITE4,(BX,BY),6,4
340 PX=PX+1:AX=AX-5
350 PO=PX+16:PP=PX+32
360 A=STRIG(CT):IFA=-1 ANDT1=0 THEN DI=1:BY=8:T1=1
```



```

370 IF DI=1 THEN BY=BY+4
380 IF BY>160 THEN DI=0:BX=AX:BY=209:T1=0
390 IF AX<-32 THEN AX=255
400 GOTO 310
410 IF BY>120 AND BX>PX+10AND BX<PX+40 THEN GOTO 430
420 RETURN
430 FOR N=1 TO 30STEP 2:CIRCLE(PO,PY),N:NEXT
440 PUTSPRITE 2,(PP+4,PY-1), 1, 2:PUTSPRITE 1,(PO-3,PY-10),1,1:PUTSPRITE 0,(PX-5
,PY+3),1,0
450 FOR N=2 TO 30STEP 2:CIRCLE(PO,PY),N,6:NEXT
460 OPEN "GRP:"AS#1:COLOR 1
470 PSET (20,20):PRINT#1,"ENHORA BUENA HAS HUNDIDO "
480 PSET (21,20):PRINT#1,"ENHORA BUENA HAS HUNDIDO "
490 PSET (60,35):PRINT#1,"EL SARATOGA
500 PSET (61,35):PRINT#1,"EL SARATOGA
510 PSET (20,50):PRINT#1,"GADAFI TE LO AGRADECERA"
520 PSET (21,50):PRINT#1,"GADAFI TE LO AGRADECERA"
530 FOR N=1 TO 5000:NEXT:RUN
540 FOR K=0 TO 4 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$(K)=B$:NEXT
550 LINE (0,140)-(255,192),4,BF
560 RETURN
570 DATA0,0,0,0,0,0,0,0,FF,31,1F,F,7,3,7,0,0,0,0,0,0,0,0,0,0,0,0,FF,89,8F,FF,FF,FF,FF,
0,0,1,7,1,3,3,0,0,FF,98,F9,FF,FF,FF,FF,0,F8,FC,6,AC,8F,FA,F8,A8,FF,6B,6B,E3,FF,F
F,FF,0,0,0,0,0,0,0,0,0,FF,92,FF,DB,FF,FF,FF,0,0,0
580 DATA0,0,0,0,0,0,FC,10,30,E0,E0,C0,C0,0
590 DATA 0,0,0,0,1F,82,5,BF,FF,BF,C,88,30,30,0,0,0,0,0,0,F0,83,87,EF,FF,FE,4,2,0
,0,0,0
600 DATA1,0,1,7,6,6,6,6,3,1,1,1,1,1,0,0,C0,80,C0,F0,B0,B0,B0,B0,E0,40,40,40,C0,C
0,80,80

```



SALTO

```

10 /*****
20 /****          ****
30 /****    SALTO    ****
40 /****          ****
50 /*****
60 /
70 /LUIS SANGUINO ARIAS


```



```

80 '
90 '      1986
100 '*****
110 '
120 CLEAR:COLOR 3,6,10
130 SCREEN 3
140 OPEN"GRP:"AS#1
150 PSET(55,80),6:PRINT#1,"SALTO"
160 FOR Y=1 TO 1500:NEXT
170 CLOSE
180 COLOR 1,2,13:SCREEN 2,2,1
190 DRAW"C15BMO,25F65R29F1D85R162U1BMO,19C7R2E8R14F28R5E24F5E6F11U8E7F39D9F9E39F
6U1E6R1D5U1E8D2F8U1E7F23D12U13"
200 PAINT (1,1),7
210 PAINT (1,191),15
220 B$="":FOR K=1 TO 3 :B$="":FOR I=1 TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT
:SPRITE$ (K)=B$:NEXT
230 OPEN"GRP:"AS#1
240 G=.05
250 SOUND 6,15:SOUND 7,7:SOUND 8,16:SOUND 9,16:SOUND 10,16:SOUND 11,0 :SOUND 12,
16:SOUND 13,0
260 ON STRIG GOSUB 370
270 X=0:Y=3:A=2:B=3
280 Y=Y+1
290 STRIG(0) ON
300 X=X+1
310 PUTSPRITE 1 ,(X,Y), 1, 1
320 PUTSPRITE 2 ,(X,Y+16),1,A
330 IF X>95 THEN Y=(Y+(1/G))
340 IF Y>155 THEN 400
350 IF Y>60 THEN A=B:GOTO 300
360 GOTO 280
370 STRIG(0) OFF
380 G=G+.1
390 RETURN
400 J=J+1
410 A(J)=X
420 IF A(J)>JJ THEN JJ=A(J)
430 LINE (60,0)-(200,10),15,BF
440 PSET(66,3),15:PRINT#1,"ULTIMO SALTO=";X
450 FOR I=1 TO 100:BEEP:NEXT
460 LINE (60,0)-(200,10),15,BF
470 PSET(70,3),7:PRINT#1,"MEJOR SALTO=";JJ
480 FOR I=1 TO 1000:NEXT
490 IF J>4 THEN 510
500 GOTO 240
510 COLOR 1,15,6
520 SCREEN 1:KEY OFF
530 LOCATE 7,10:PRINT"TU MEJOR SALTO"
540 LOCATE 5,13:PRINT"HA SIDO=";JJ;"METROS"
550 LOCATE 1,20:PRINT"OTRA PARTIDA SI O NO [S/N]"
560 X$=INKEY$
570 IF X$="S" OR X$="s" THEN RUN
580 IF X$="n" OR X$="N" THEN CLS:SCREEN 0:END
590 GOTO 560
600 DATA0,1,3,3,3,1,3,7,7,7,7,7,3,3,7,7,80,C0,C0,E0,C0,80,80,C0,C0,C0,E0,B0,98,c
,84,88,7,7,1,0,21,12,8,5,3,1,0,0,0,0,0,0,90,E0,E0,F0,30,60,C0,80,80,80,80,40,20,
10,C,0
610 DATA7,7,3,3,3,3,6,6,C,C,1C,1E,FF,0,0,0,88,88,8,8,8,8,8,8,1C,8,0,4,F8,0,0,0

```

RUTA-X

```

10 / *****
20 / *****
30 / ***** RUTA-X *****
40 / *****
50 / *****
60 /
70 / LUIS SANGUINO ARIAS
80 /
90 / 1986
100 / *****
110 /
120 COLOR 1,4,6
130 SCREEN 1,3:KEY OFF
140 GOSUB 370
150 G=1:A=1
160 PUT SPRITE 4,(83,82),1,4
170 LOCATE 13,12:PRINT"RUTA-X"
180 LOCATE 2,22:PRINT"LUIS SANGUINO ARIAS - 1986"
190 FOR I= 1 TO 500:A=A+1:COLOR A-1,A+1,A
200 IF A>13 THEN A=1
210 NEXT
220 PUT SPRITE 4,(83,209),1,4
230 CLS
240 COLOR 1,14
250 LOCATE 3,10:PRINT"JOYSTICK O CURSOR [J/C]"
260 X$=INKEY$
270 IF X$="j" OR X$="J" THEN CT=1:GOTO 330
280 IF X$="c" OR X$="C" THEN CT=0:GOTO 330
290 R=R+1
300 IF R>14 THEN R=1
310 VDP(7)=R
320 GOTO 260
330 OPEN"GRP:"AS 1
340 COLOR 10,10,1
350 SCREEN 2,2
360 RESTORE
370 FOR K=1 TO 4 :B$="":FORI=1TO 32 :READ A$:B$=B$+CHR$(VAL("&H"+A$)):NEXT:SPRIT
E$(K)=B$:NEXT
380 IF G=0 THEN RETURN


```




```




390 B$="":FOR I=1 TO 3:READ A:B$=B$+CHR$(A):NEXT:SPRITE$(6)=B$
400 LINE (0,67)-(255,125),4,BF
410 LINE (0,182)-(90,192),1,BF
420 PSET(1,184),1:PRINT#1,"PUNTOS=";PT
430 TIME =0
440 ON SPRITE GOSUB 700
450 X=255
460 R=R+8:IEX=IEX-10:SEX=SEX-8
470 IF TIME>4000 THEN 800
480 TY=87:CS=209:CI=209
490 BX=BX+7 :SY=SY-K
500 LINE (X-R,94)-(X-30-R,98),10,BF
510 IF X-R<-6 THEN R=0:GOTO 450
520 SPRITE ON
530 LINE (X-1-R,94)-(X+6-R,98),4,BF
540 D=STICK(CT)
550 IF D=1 THEN TY=209:CS=87:PUT SPRITE4,(36,TY)
560 IF D=5 THEN TY=209:CI=87:PUT SPRITE4,(36,TY)
570 PUTSPRITE 3 ,(IEX,150),6,3
580 PUTSPRITE 1 ,(36,CS),1,1
590 PUTSPRITE 5 ,(SEX,20),4,3
600 PUTSPRITE 2 ,(36,CI),1,2
610 PUTSPRITE 4 ,(36,TY),1,4
620 IF DS=1 THEN 660
630 IF DI=1 THEN 660
640 IF STRIG(CT)=-1 AND CS=87 THEN SY=91:BX=42:K=7:DS=1
650 IF STRIG(CT)=-1 AND CI=87 THEN SY=103:BX=48:K=-7:DI=1 ELSE GOTO 690
660 PUT SPRITE 6,(BX,SY),1,6
670 IF SY<0 THEN DS=0:DI=0
680 IF SY>193 THEN DI=0:DS=0
690 GOTO 460
700 SPRITE OFF:PT=PT+50:LINE(50,182)-(90,192),1,BF:PSET(45,184),1:PRINT#1,PT
710 SOUND 6,15:SOUND 7,7:SOUND 8,16:SOUND 9,16:SOUND 10,16:SOUND 11,0 :SOUND 12,
16:SOUND 13,0
720 IF DS=1 THEN SEX=255
730 IF DI=1 THEN IEX=255
740 DS=0:DI=0:SY=209
750 PUT SPRITE 6,(BX,SY),1,6
760 RETURN
770 DATADB,DB,DB,DB,84,47,5C,37,37,5C,47,84,DB,DB,DB,DB,68,68,69,6B,16,EC,58,70,
60,40,E0,10,68,68,68,68,DB,DB,DB,DB,84,47,5C,37,37,5C,47,84,DB,DB,DB,DB,68,68,68
,68,10,E0,40,60,70,58,EC,16,6B,69,68,68,0,D,D,ED,ED,FF,92,9D,9D,92,FF,ED,ED,D,D,
0,0,B6
780 DATAB6,B7,B5,F9,16,EA,EA,16,F9,B5,B7,B6,B6,0
790 DATADB,DB,DB,DB,84,47,5C,37,37,5C,47,84,DB,DB,DB,DB,68,68,68,68,10,E0,40,7F,
7F,40,E0,10,68,68,68,68,2,7,2
800 COLOR 1,14,6
810 SCREEN 1
820 LOCATE 1,10:PRINT"OTRA PARTIDA SI O NO [S/N]
830 LOCATE 1,4:PRINT"TUS PUNTOS HAN SIDO =";PT
840 X$=INKEY$
850 VDP(7)=I
860 IF X$="S" OR X$="s" THEN CLEAR:G=1:GOTO 230
870 IF X$="n" OR X$="N" THEN END
880 I=I+1
890 IF I>15 THEN I=0
900 GOTO 840

```





SUSCRI-
BETE
LLAMAN
DO A



BIBLIOGRAFIA

Para Programadores MSX tu fiel compañero.

Esto no es un libro para el principiante novato, al contrario, introduce al programador MSX "dentro" del ordenador, para mostrarle exactamente como funciona y como obtener el máximo partido.

Los autores han dividido su libro en dos partes.


La primera parte cubre el diseño del sistema, como el vocabulario de BASIC se relaciona con los ordenadores MSX, y una introducción código máquina. Encontrará explicaciones detalladas sobre la organización de la memoria, los modos de pantalla y los circuitos de vídeo y sonido.

La segunda parte se dedica al uso del lenguaje ensamblador en el sistema MSX. Los principales apartados incluyen los puntos:

- El procesador de pantalla de vídeo.
- El circuito de sonido AY-3-8910.
- Entradas/Salidas.

Tanto si necesitas escribir programas de alta eficacia o solamente quieres como trabaja tu ordenador, este libro será inestimable. Pensamos que rápidamente se convertirá en el texto base para aquellos programadores que quieren hacer algo más que escribir programas en BASIC.

P.V.P. 1800 pts!



CODIGO MAQUINA

La llegada de standar MSX, marca un paso significativo dentro del mundo de los ordenadores domésticos. Ofrece infinidad de excitantes oportunidades para programadores y usuarios.


Este libro, supone que no tienes conocimientos previos sobre código máquina. Explicando las equivalentes en código máquina, de las principales instrucciones en Basic, tales como IF, FOR/NEXT, PRINT, GOTO, GOSUB, etc... Continúa con la descripción muy detallada de las rutinas individuales de un simple juego: "INVASOR DEL ESPACIO", y cómo se enlazan estas rutinas que puedes aprovechar en tus programas en Basic.

- Aprende lo fácil que es incorporar rutinas en código máquina dentro de tus programas en Basic.
- Usa el código máquina para crear sonido y gráficos.
- Diseña tus propios caracteres y "escenarios" con dos programas fáciles de usar.

Tanto si simplemente desea incluir rutinas para acelerar y mejorar tus programas en Basic, como si quieres un programa completo y potente en código máquina, encontrarás este libro extremadamente útil. En él está suprimida la teoría que dificulta el código máquina, y se presenta de una forma práctica y divertida de aprenderlo.

P.V.P. 1.200 pts.

PROGRAMACION PRACTICA



tca-ma

TARJETA DE PEDIDO

--	--	--	--	--	--	--	--

Domicilio de envíos:

Nombre y apellidos
 Domicilio N.º Piso Pta.
 C. P. Ciudad Provincia

Ruego sírvanse remitirme CONTRA REEMBOLSO los siguientes libros:

Número	Cantidad	TITULO Y AUTOR	Importe

Fecha / /

Fdo.:

• La mayor variedad de libros de microinformática, capaces de satisfacer todas sus necesidades, ya sean profesionales, familiares, culturales...

• Todo tipo de información bibliográfica sobre microordenadores, desde AMSTRAD a Sinclair QL

Libros, Revistas, Suscripciones, Importación y Distribución
 Chiquinquira, 28. Local 4 (Cocuy). 28033 MADRID.
 Telef.: 764 50 95


DISPONIBLE PARA ZX SPECTRUM

SOFTWARE

Sound-on-Sound

Sound on Sound es una marca registrada
producida y distribuida por Iberofón, S. A.
Telf. 671.22.00 / 04/08/12/16

JUEGA CON EL FUTURO



¡¡¡NO LO SUEÑES!!! ¡JUEGALO!
SIENTE LA EMOCION DE LO DESCONOCIDO
CORRE TU PROPIO RIESGO

SALVA A TU COMPAÑERO/A ATRAPADO/A
REUNE LOS FRAGMENTOS DEL CUADRO
SON TU AMULETO

"BLACK JACK" ¡¡¡EL JUEGO DE CARTAS
MAS EMOCIONANTE!!!

¡¡¡CASINO ROYAL!!!
"LA EMOCION DEL JUEGO POR TODO LO ALTO"

¡¡¡POR FIN EN CASTELLANO!!!
LA PRIMERA COMEDIA MUSICAL EN VIDEO-JUEGO

TODO SOBRE EL

AMSTRAD

AÑO II • N.º 9

SUMARIO:

- NEVERA
- PASILLO MORTAL
- TRANS-SPAIN
- BI-LASER
- CHARCA
- TRAGABOLAS
- CAFE
- CUCO
- CUERPO HUMANO

420 Ptas.

10
PROGRAMAS



¡ya está a la venta!